SSC-N-88

S. Peggs
Dec. 5, 1985

An example of a standard input lattice file

INTRODUCTION

The main body of this note is a lattice file representing the late separation luminosity optics LCW005[1], as it has been developed in standard input format[2,3]. However, the goal of the note is NOT to introduce the reader to a particular lattice, but rather to present the standard format as a natural readable form in which to record lattices. It is the form, rather than the content, of the lattice file which is interesting here. The contention is that readers who have little or no detailed knowledge of either late separation optics, or of the standard format, will quickly and easily be able to understand what is meant, and how the lattice is constructed.

It is not sufficient merely to use the standard format in order to ensure that a lattice file is easy to understand. The file must also include a minimal number of comments, should be well organised, and should conform at least loosely to a set of informal naming conventions, such as using Q for the first letter in the name of all quadrupoles. While the standard input format has a practical definition - will DIMAT (for example) accept the file? - the construction of the informal naming conventions is largely a matter of taste, and is certainly almost impossible to standardise, or even to define rigorously.

This note should be accepted as the casual document which it is intended to be, merely providing food for thought on the subjects of the standard input format, and the ergonomic arrangement of the file.

Finally, the reader might find it more logical to understand the lattice by starting at the end of the file and working backwards, that is, starting from the lattice as a whole, and breaking it down successively into its substructures.

ONE SET OF INFORMAL NAME CONVENTIONS

At the risk of confusing the issue, and weakening the claim that the standard input format is so natural to understand that no prior knowledge is needed, the following self imposed naming conventions have been used more or less rigorously. The alphabetically organised list of these conventions should be freely ignored, especially at a first pass!

B	first letter of a bend, e.g. B
CDF, CFD	a half cell, from the center of a D to an F quadrupole, etc., e.g. CDF9
CELLF, CELLD	a complete cell, between centers of F quads, etc., e.g. CELLFE6
CT	associates an element with a constant phase trombone, e.g. MCT1
D	first letter of a drift, e.g. DSPOOL
	OR, defocussing, e.g. QD6
	OR, implies a downward bend or step, e.g. LBND9
F	focussing, e.g. CELLF6
K	implies a magnet strength, e.g. K19
L	first letter of a length, e.g. LQUAD
LB	low beta matching line, e.g. LBXU9
M	first letter of a marker, e.g. MF6
N	indicates entering an element, or line, e.g. QFNPT
PT	associates an object with a phase trombone, e.g. MPT1
Q	first letter of a quadrupole, e.g. QF6
SEP	vertical beam separation line, e.g. SEPND6
T	first letter of a quadrupole thickness, e.g. T1
U	associates an object with a utility straight, e.g. DU16
	OR, associates an upward vertical step with an object, e.g. SEPXU6
V	vertical bend or step, e.g. V2UP
X	indicates exiting an element, or line, e.g. LBXU6
6	associates 60 degree phase advance with an object, e.g. CELLF6
9	associates 90 degree phase advance with an object, e.g. LBXU9

REFERENCES

1	SSC-N-87,	Oct. 1985,	Closed luminosity and injection optics in a 'late' separation scheme for the SSC
2	Snowmass,	1984, p 389,	A standard input language for particle beam and accelerator computer programs
3	SSC-TR-2001,	1984, p 125,	Report of the group on a common input format

```
TITLE
  LCW005 -- 6.6T, luminosity optics, bstar = 0.5 m.

!**********************************
!FUNDAMENTALS

PARA, PI = 3.1415926535897932;          PARA, CLIGHT = 0.2997925
PARA, ENERGY = 20000.0;  PARA, BRHO = ENERGY / CLIGHT
PARA, LHALF = 96.0;  PARA, LEND = 0.375;  PARA, LQ = 3.30;  PARA, LD = 16.6
PARA, LQUAD = LQ + 2.0*LEND;   PARA, LSLOT = LD + 2.0*LEND
PARA, LSPOOL = LHALF - LQUAD - 5.0*LSLOT
PARA, LSEXT = LSPOOL - 2.0*LEND
PARA, NDARC = 3610; PARA, NDPT9 = 126; PARA, NDSEP6 = 40; PARA, NHDS = 48
PARA, BDSFACT = 0.8778
PARA, NDIPOLES = NDARC + NDPT9 + NDSEP6 + NHDS*BDSFACT
PARA, THETA = 2.0*PI / NDIPOLES; PARA, RHO = LD / THETA
PARA,  BFIELD = BRHO / RHO
VALUE, NDIPOLES;  VALUE, THETA;  VALUE, BFIELD
!There must be an odd number of half cells in the each of the 2 arcs,
!with 5 dipoles per half cell, so there are 3610/10 = 361 HALF cells per arc.

!**********************************
!DRIFTS: MISCELLANEOUS

DSLOT:    DRIFT,  L = LSLOT
DSPOOL:   DRIFT,  L = LSPOOL
DEND:     DRIFT,  L = LEND

!**********************************
!DRIFTS: LOW BETAS AND INTERSECTION REGIONS

! Tn => quad n Thickness,  Ln => (n-1)-to-(n) quad spacing
PARA, T1 = 13.6;  PARA, T2 = 21.8;  PARA, T3 = 12.1;  PARA, T4 = 9.0
PARA, L1 = 27.3;  PARA, L2 = 19.7;  PARA, L3 = 18.45
PARA, L49 = 195 - L1 - L2 - L3;    PARA, L59 = 50.0
PARA, L46 = 210 - L1 - L2 - L3;    PARA, L56 = 51.0
PARA, L6 = 90.0
D1:    DRIFT,  L = 0.1*(L1 - 0.5*T1)
D2:    DRIFT,  L = L2 - 0.5*(T1 + T2)
D3:    DRIFT,  L = L3 - 0.5*(T2 + T3)
D49:   DRIFT,  L = 0.1*(L49 - 0.5*(T3 + T4))
D46:   DRIFT,  L = 0.1*(L46 - 0.5*(T3 + T4))
D59:   DRIFT,  L = L59 - 0.5*(T4 + LVSQUAD)
D56:   DRIFT,  L = L56 - 0.5*(T4 + LQUAD)

!**********************************
!DRIFTS: CELLS AND HALF CELLS

!separator
PARA, LVSHALF = 90.0;   PARA, LVSQ = 5.0;   PARA, LVSQUAD = LVSQ + 2*LEND
PARA, BV1 = 6.7;  PARA, LV1 = 41.0
PARA, BV2 = 6.7;  PARA, LV2 = 41.0
PARA,  LVS1 = L6 - LVSQUAD - 2*LV1 - 4*LEND
PARA,  LVS2 = LVSHALF - LVSQUAD - 2*LV2 - 4*LEND
PARA,  LVSE9 = LVSHALF - LVSQUAD;   PARA,  LVSE6 = LHALF - LQUAD
DVS1:  DRIFT,  L = LVS1;             DVS2:  DRIFT,  L = LVS2
```

```
DDVS1: DRIFT,  L = LVSQUAD - LQUAD;  DDVS2: DRIFT,  L = LVSE6 - LVSE9
DVSE9: DRIFT,  L = LVSE9;            DVSE6: DRIFT,  L = LVSE6
!horizontal dispersion matching cell
PARA, LST = 9.48;   PARA, LFIN = LVSHALF - LVSQUAD - 2*LSLOT - LST
DST:    DRIFT, L = LST;    DFIN:    DRIFT, L = LFIN
!phase trombones
PARA, LPTHALF = 70.0;  PARA, LPTQ = 7.0;  PARA, LPTQUAD = LPTQ + 2*LEND
PARA, LSPLPT = LPTHALF - LPTQUAD - 3*LSLOT
DSP2:    DRIFT, L = 0.5*LSPLPT
!join between utilities
PARA, LJHALF = 110.0;  PARA, LJQ = 4.2;  PARA, LJQUAD = LJQ + 2*LEND
PARA, L4J = LJHALF - 4*LSLOT - LJQUAD
PARA, L5NJ = 0.938;     PARA, L5XJ = LJHALF - 5*LSLOT - LJQUAD - L5NJ
PARA, LEJ = LJHALF - LJQUAD
D4J:   DRIFT, L = L4J
D5NJ:  DRIFT, L = L5NJ;   D5XJ:  DRIFT, L = L5XJ
DEJ:   DRIFT, L = LEJ

!***********************************
!DRIFTS: UTILITIES

! Tn => quad n Thickness,  Ln => (n-1)-to-(n) quad spacing
PARA, LXFER = 65.0
!90 degree utilities
PARA, LTOTU9 = L1 + L2 + L3 + L49 + L59 + L6 + LVSHALF + LXFER
PARA, TU19 = 7.0; PARA, TU29 = 7.0; PARA, TU39 = LQ;  PARA, TU49 = 0.5*LVSQ
PARA, LU19 = 250.0; PARA, LU29 = 10.0;  PARA, LU39 = 200.0
PARA, LU49 = LTOTU9 - LU19 - LU29 - LU39
DU19:   DRIFT,  L = 0.1*(LU19 - 0.5*TU19)
DU29:   DRIFT,  L = LU29 - 0.5*(TU19 + TU29)
DU39:   DRIFT,  L = 0.1*(LU39 - 0.5*(TU29 + TU39))
DU49:   DRIFT,  L = LU49 - 0.5*TU39 - TU49
!60 degree utilities
PARA, LTOTU6 = L1 + L2 + L3 + L46 + L56 + L6 + 2*LHALF + LXFER
PARA, TU16 = 7.0; PARA, TU26 = 7.0; PARA, TU36 = LQ;  PARA, TU46 = LQ
PARA, LU16 = 250.0; PARA, LU26 = 15.0;  PARA, LU36 = 200.0;  PARA, LU46 = 30.0
PARA, LU56 = LTOTU6 - LU16 - LU26 - LU36 - LU46
DU16:   DRIFT,  L = 0.1*(LU16 - 0.5*TU16)
DU26:   DRIFT,  L = LU26 - 0.5*(TU16 + TU26)
DU36:   DRIFT,  L = 0.1*(LU36 - 0.5*(TU26 + TU36))
DU46:   DRIFT,  L = LU46 - 0.5*(TU36 + TU46)
DU56:   DRIFT,  L = LU56 - 0.5*(TU46 + LQUAD)
!join utilities
PARA, LTOTUJ = L1+L2+L3+L49+L59+L6 + 4*LVSHALF + 3.5*(LPTHALF - LJHALF) - LXFER
PARA, TU1J = 7.0;  PARA, TU2J = 7.0;  PARA, TU3J = LQ;  PARA, TU4J = 0.5*LJQ
PARA, LU1J = 250.0;  PARA, LU2J = 15.0;  PARA, LU3J = 200.0
!PARA, LU4J = LTOTUJ - LU1J - LU2J - LU3J
PARA, LU4J = LTOTUJ - LU1J - LU2J - LU3J
DU1J:   DRIFT,  L = 0.1*(LU1J - 0.5*TU1J)
DU2J:   DRIFT,  L = LU2J - 0.5*(TU1J + TU2J)
DU3J:   DRIFT,  L = 0.1*(LU3J - 0.5*(TU2J + TU3J))
DU4J:   DRIFT,  L = LU4J - 0.5*TU3J - TU4J

!***********************************
!BENDS: HORIZONTAL
```

```
!regular bends in main arc
BFULL:    SBEND, L = LD,    ANGLE = THETA
B:        LINE = (DEND,BFULL,DEND)
!full bend, with spool replaced by a drift, in 60 deg. vertical separator
BVSF6:     LINE = (5*B,DSPOOL)
!horizontal dispersion matcher/suppressor in 90 degree vertical separator
BDISP:    SBEND, L = LD,    ANGLE = BDSFACT * THETA
BDS:    LINE = (DEND,BDISP,DEND)
HDSN9:  LINE = (DST,2*BDS,DFIN);   HDSX9:   LINE = (DFIN,2*BDS,DST)
!symmetric phase trombone bend
BPT:    LINE = (DSP2,3*B,DSP2)
!join between utilities
PARA, THETAJ = ((21 + 8*BDSFACT)*THETA)/28
BJBEND:  SBEND, L = LD, ANGLE = THETAJ
BJ:    LINE = (DEND,BJBEND,DEND)
B5NJ:  LINE = (D5NJ,5*BJ,D5XJ)
B5XJ:  LINE = (D5XJ,5*BJ,D5NJ)
B4J:   LINE = (2*BJ,D4J,2*BJ)

!************************************
!BENDS: VERTICAL

!first vertical step
V1UP:     SBEND, L = LV1,     ANGLE =-BV1 * LV1/BRHO,    TILT
V1DOWN:   SBEND, L = LV1,     ANGLE = BV1 * LV1/BRHO,    TILT
V1U:   LINE = (DEND,V1UP,DEND);      V1D:   LINE = (DEND,V2DOWN,DEND)
VSU19:    LINE = (V1U,DVS1,V1D);    VSU16:    LINE = (DDVS1,VSU19)
VSD19:    LINE = (V1D,DVS1,V1U);    VSD16:    LINE = (VSD19,DDVS1)
!second vertical step
V2UP:     SBEND, L = LV2,     ANGLE =-BV2 * LV2/BRHO,      TILT
V2DOWN:   SBEND, L = LV2,     ANGLE = BV2 * LV2/BRHO,      TILT
V2U:   LINE = (DEND,V2UP,DEND);      V2D:   LINE = (DEND,V2DOWN,DEND)
VSU29:    LINE = (V2U,DVS2,V2D);    VSU26:    LINE = (DDVS2,VSU29)
VSD29:    LINE = (V2D,DVS2,V2U);    VSD26:    LINE = (VSD29,DDVS2)

!************************************
!QUADS: CELLS AND HALF CELLS

!60 degrees
PARA, DFK6 = 0.0;  PARA, DDK6 = 0.0
PARA, FK6 = 0.00317579 + DFK6;      PARA, DK6 = 0.00317637 + DDK6
QFHALF60: QUAD, L = 0.5*LQ, K1 = FK6;  QDHALF60: QUAD, L = 0.5*LQ, K1 = -DK6
QFN6: LINE = (DEND,QFHALF60);      QFX6: LINE = (QFHALF60,DEND)
QDN6: LINE = (DEND,QDHALF60);      QDX6: LINE = (QDHALF60,DEND)
MQF6:  MARK;  MQD6:  MARK
QF6: LINE = (QFN6,MQF6,QFX6);    QD6: LINE = (QDN6,MQD6,QDX6)
!90 degree separator quads
PARA, DFK9 = 0.0;  PARA, DDK9 = 0.0
PARA, FK9 = 0.00320268 + DFK9;      PARA, DK9 = 0.00320267 + DDK9
QFHALF90: QUAD, L=0.5*LVSQ, K1=FK9;  QDHALF90: QUAD, L=0.5*LVSQ, K1=-DK9
QFN9: LINE = (DEND,QFHALF90);      QFX9: LINE = (QFHALF90,DEND)
QDN9: LINE = (DEND,QDHALF90);      QDX9: LINE = (QDHALF90,DEND)
MQF9:  MARK;  MQD9:  MARK
QF9: LINE = (QFN9,MQF9,QFX9);    QD9: LINE = (QDN9,MQD9,QDX9)
!join between utilities
PARA, DFKJ = 0.0;    PARA, DDKJ = 0.0
```

```
PARA, FKJ = 0.00310083 + DFKJ;   PARA, DKJ = 0.00310084 + DDKJ
QFHALFJ: QUAD, L =0.5*LJQ, K1=FKJ;    QDHALFJ: QUAD, L=0.5*LJQ, K1=-DKJ
QFNJ: LINE = (DEND,QFHALFJ);   QFXJ: LINE = (QFHALFJ,DEND)
QDNJ: LINE = (DEND,QDHALFJ);   QDXJ: LINE = (QDHALFJ,DEND)
MQFJ:  MARK;  MQDJ:  MARK
QFJ:  LINE = (QFNJ,MQFJ,QFXJ);    QDJ:  LINE = (QDNJ,MQDJ,QDXJ)
!neutral phase trombone
PARA, DFKPT = 0.0;  PARA, DDKPT = 0.0
PARA, FKPT = 0.00298771 + DFKPT;   PARA, DKPT = 0.00298777 + DDKPT
QFHALFPT: QUAD, L=0.5*LPTQ, K1=FKPT;   QDHALFPT: QUAD, L=0.5*LPTQ, K1=-DKPT
QFNPT: LINE = (DEND,QFHALFPT);   QFXPT: LINE = (QFHALFPT,DEND)
QDNPT: LINE = (DEND,QDHALFPT);   QDXPT: LINE = (QDHALFPT,DEND)
MQFPT: MARK;   MQDPT: MARK
QFPT: LINE = (QFNPT,MQFPT,QFXPT);    QDPT: LINE = (QDNPT,MQDPT,QDXPT)

!**********************************
!QUADS: PHASE TROMBONES

!segment phase trombone
PARA, DG1 = 0.0;  PARA, DG2 = 0.0;  PARA, DG3 = 0.0;  PARA, DG4 = 0.0
PARA, G1 = 0.97358985 + DG1;   PARA, G8 = G1
PARA, G2 = 0.99061721 + DG2;   PARA, G7 = G2
PARA, G3 = 0.99129816 + DG3;   PARA, G6 = G3
PARA, G4 = 0.98725322 + DG4;   PARA, G5 = G4
P1: QUAD, L=0.5*LPTQ, K1= -G1*DKPT;      P2: QUAD, L=0.5*LPTQ, K1= G2*FKPT
P3: QUAD, L=0.5*LPTQ, K1= -G3*DKPT;      P4: QUAD, L=0.5*LPTQ, K1= G4*FKPT
P5: QUAD, L=0.5*LPTQ, K1= -G5*DKPT;      P6: QUAD, L=0.5*LPTQ, K1= G6*FKPT
P7: QUAD, L=0.5*LPTQ, K1= -G7*DKPT;      P8: QUAD, L=0.5*LPTQ, K1= G8*FKPT
MPT1:  MARK; MPT2:  MARK; MPT3:  MARK; MPT4:  MARK
MPT5:  MARK; MPT6:  MARK; MPT7:  MARK; MPT8:  MARK
PTD1:  LINE=(MPT1,P1,DEND);          PTF2:  LINE=(DEND,P2,MPT2,P2,DEND)
PTD3:  LINE=(DEND,P3,MPT3,P3,DEND);  PTF4:  LINE=(DEND,P4,MPT4,P4,DEND)
PTD5:  LINE=(DEND,P5,MPT5,P5,DEND);  PTF6:  LINE=(DEND,P6,MPT6,P6,DEND)
PTD7:  LINE=(DEND,P7,MPT7,P7,DEND);  PTF8:  LINE=(DEND,P8,MPT8)
!constant phase trombone
PARA, G1C = 1.23994693 ;   PARA, G8C = G1C
PARA, G2C = 1.12589868 ;   PARA, G7C = G2C
PARA, G3C = 1.24561792 ;   PARA, G6C = G3C
PARA, G4C = 1.23518552 ;   PARA, G5C = G4C
C1: QUAD, L=0.5*LPTQ, K1= -G1C*DKPT;   C2: QUAD, L=0.5*LPTQ, K1= G2C*FKPT
C3: QUAD, L=0.5*LPTQ, K1= -G3C*DKPT;   C4: QUAD, L=0.5*LPTQ, K1= G4C*FKPT
C5: QUAD, L=0.5*LPTQ, K1= -G5C*DKPT;   C6: QUAD, L=0.5*LPTQ, K1= G6C*FKPT
C7: QUAD, L=0.5*LPTQ, K1= -G7C*DKPT;   C8: QUAD, L=0.5*LPTQ, K1= G8C*FKPT
MCT1:  MARK; MCT2:  MARK; MCT3:  MARK; MCT4:  MARK
MCT5:  MARK; MCT6:  MARK; MCT7:  MARK; MCT8:  MARK
CTD1:  LINE=(MCT1,C1,DEND);          CTF2:  LINE=(DEND,C2,MCT2,C2,DEND)
CTD3:  LINE=(DEND,C3,MCT3,C3,DEND);  CTF4:  LINE=(DEND,C4,MCT4,C4,DEND)
CTD5:  LINE=(DEND,C5,MCT5,C5,DEND);  CTF6:  LINE=(DEND,C6,MCT6,C6,DEND)
CTD7:  LINE=(DEND,C7,MCT7,C7,DEND);  CTF8:  LINE=(DEND,C8,MCT8)

!**********************************
!QUADS: LOW BETA

!90 degree low beta
PARA, DK19=0.0;  PARA, DK29=0.0;  PARA, DK39=0.0;  PARA, DK49 = 0.0
PARA, K19 = 0.00350454 + DK19;    PARA, K29 = 0.00363570 + DK29
```

```
PARA, K39 = 0.00387570 + DK39;     PARA, K49 = 0.00349012 + DK49
RX19: QUAD, L = T1/2.,   K1 =-K19;     RX29: QUAD, L = T2/4.,   K1 = K29
RX39: QUAD, L = T3/2.,   K1 =-K39;     RX49: QUAD, L = T4/2.,   K1 = K49
RN19: QUAD, L = T1/2.,   K1 = K19;     RN29: QUAD, L = T2/4.,   K1 =-K29
RN39: QUAD, L = T3/2.,   K1 = K39;     RN49: QUAD, L = T4/2.,   K1 =-K49
MQ19:  MARK; MQ29:  MARK; MQ39:  MARK; MQ49:  MARK
QX19:  LINE = (MQ19,2*(RX19,MQ19)); QX29:  LINE = (MQ29,4*(RX29,MQ29))
QX39:  LINE = (MQ39,2*(RX39,MQ39)); QX49:  LINE = (MQ49,2*(RX49,MQ49))
QN19:  LINE = (MQ19,2*(RN19,MQ19)); QN29:  LINE = (MQ29,4*(RN29,MQ29))
QN39:  LINE = (MQ39,2*(RN39,MQ39)); QN49:  LINE = (MQ49,2*(RN49,MQ49))
!60 degree low beta
PARA, DK16=0.0;  PARA, DK26=0.0;  PARA, DK36=0.0;  PARA, DK46 = 0.0
PARA, K16 = 0.00359594 + DK16;     PARA, K26 = 0.00360845 + DK26
PARA, K36 = 0.00376545 + DK36;     PARA, K46 = 0.00316456 + DK46
RX16: QUAD, L = T1/2.,   K1 =-K16;     RX26: QUAD, L = T2/4.,   K1 = K26
RX36: QUAD, L = T3/2.,   K1 =-K36;     RX46: QUAD, L = T4/2.,   K1 = K46
RN16: QUAD, L = T1/2.,   K1 = K16;     RN26: QUAD, L = T2/4.,   K1 =-K26
RN36: QUAD, L = T3/2.,   K1 = K36;     RN46: QUAD, L = T4/2.,   K1 =-K46
MQ16:  MARK; MQ26:  MARK; MQ36:  MARK; MQ46:  MARK
QX16:  LINE = (MQ16,2*(RX16,MQ16)); QX26:  LINE = (MQ26,4*(RX26,MQ26))
QX36:  LINE = (MQ36,2*(RX36,MQ36)); QX46:  LINE = (MQ46,2*(RX46,MQ46))
QN16:  LINE = (MQ16,2*(RN16,MQ16)); QN26:  LINE = (MQ26,4*(RN26,MQ26))
QN36:  LINE = (MQ36,2*(RN36,MQ36)); QN46:  LINE = (MQ46,2*(RN46,MQ46))
!join low beta
PARA, DK1J=0.0;  PARA, DK2J=0.0;  PARA, DK3J=0.0;  PARA, DK4J = 0.0
PARA, K1J = 0.00357823 + DK1J;     PARA, K2J = 0.00364524 + DK2J
PARA, K3J = 0.00386796 + DK3J;     PARA, K4J = 0.00358534 + DK4J
RX1J: QUAD, L = T1/2.,   K1 =-K1J;     RX2J: QUAD, L = T2/4.,   K1 = K2J
RX3J: QUAD, L = T3/2.,   K1 =-K3J;     RX4J: QUAD, L = T4/2.,   K1 = K4J
RN1J: QUAD, L = T1/2.,   K1 = K1J;     RN2J: QUAD, L = T2/4.,   K1 =-K2J
RN3J: QUAD, L = T3/2.,   K1 = K3J;     RN4J: QUAD, L = T4/2.,   K1 =-K4J
MQ1J:  MARK; MQ2J:  MARK; MQ3J:  MARK; MQ4J:  MARK
QX1J:  LINE = (MQ1J,2*(RX1J,MQ1J)); QX2J:  LINE = (MQ2J,4*(RX2J,MQ2J))
QX3J:  LINE = (MQ3J,2*(RX3J,MQ3J)); QX4J:  LINE = (MQ4J,2*(RX4J,MQ4J))
QN1J:  LINE = (MQ1J,2*(RN1J,MQ1J)); QN2J:  LINE = (MQ2J,4*(RN2J,MQ2J))
QN3J:  LINE = (MQ3J,2*(RN3J,MQ3J)); QN4J:  LINE = (MQ4J,2*(RN4J,MQ4J))

!***********************************
!QUADS: UTILITIES

!90 degree utility
PARA, DKU19=0.0;  PARA, DKU29=0.0;  PARA, DKU39=0.0;  PARA, DKU49=0.0
PARA, KU19 =  0.00369357 + DKU19;  PARA, KU29 = 0.00386472 + DKU29
PARA, KU39 = -0.00229390 + DKU39;  PARA, KU49 = 0.00245781 + DKU49
QUX19: QUAD, L = 0.5*TU19, K1 = KU19;  QUN19: QUAD, L = 0.5*TU19, K1 =-KU19
QUX29: QUAD, L = 0.5*TU29, K1 =-KU29;  QUN29: QUAD, L = 0.5*TU29, K1 = KU29
QUX39: QUAD, L = 0.5*TU39, K1 = KU39;  QUN39: QUAD, L = 0.5*TU39, K1 =-KU39
QUX49: QUAD, L = TU49,     K1 = KU49;  QUN49: QUAD, L = TU49,     K1 =-KU49
MU19: MARK; MU29: MARK; MU39: MARK; MU49: MARK
QUF19:  LINE = (QUX19,MU19,QUX19);    QUD19:  LINE = (QUN19,MU19,QUN19)
QUD29:  LINE = (QUX29,MU29,QUX29);    QUF29:  LINE = (QUN29,MU29,QUN29)
QUF39:  LINE = (QUX39,MU39,QUX39);    QUD39:  LINE = (QUN39,MU39,QUN39)
QUF49:  LINE = (QUX49,MU49);          QUD49:  LINE = (MU49,QUN49)
!60 degree utility
PARA, DKU16=0.0;  PARA, DKU26=0.0;  PARA, DKU36=0.0;  PARA, DKU46=0.0
PARA, KU16 =  0.00258402 + DKU16; PARA, KU26 = 0.00275798 + DKU26
```

```
PARA, KU36 = -0.00074590 + DKU36; PARA, KU46 = 0.00240875 + DKU46
QUX16: QUAD, L = 0.5*TU16, K1 = KU16;  QUN16: QUAD, L = 0.5*TU16, K1 =-KU16
QUX26: QUAD, L = 0.5*TU26, K1 =-KU26;  QUN26: QUAD, L = 0.5*TU26, K1 = KU26
QUX36: QUAD, L = 0.5*TU36, K1 = KU36;  QUN36: QUAD, L = 0.5*TU36, K1 =-KU36
QUX46: QUAD, L = TU46,     K1 = KU46;  QUN46: QUAD, L = TU46,     K1 =-KU46
MU16: MARK; MU26: MARK; MU36: MARK; MU46: MARK
QUF16:  LINE = (QUX16,MU16,QUX16);     QUD16:  LINE = (QUN16,MU16,QUN16)
QUD26:  LINE = (QUX26,MU26,QUX26);     QUF26:  LINE = (QUN26,MU26,QUN26)
QUF36:  LINE = (QUX36,MU36,QUX36);     QUD36:  LINE = (QUN36,MU36,QUN36)
QUF46:  LINE = (QUX46,MU46);           QUD46:  LINE = (MU46,QUN46)
!join utility
PARA, DKU1J=0.0;  PARA, DKU2J=0.0;  PARA, DKU3J=0.0;  PARA, DKU4J=0.0
PARA, KU1J = 0.00271589 + DKU1J; PARA, KU2J = 0.00291503 + DKU2J
PARA, KU3J = -0.00162719 + DKU3J; PARA, KU4J = 0.00258056 + DKU4J
QUX1J: QUAD, L = 0.5*TU1J, K1 = KU1J;  QUN1J: QUAD, L = 0.5*TU1J, K1 =-KU1J
QUX2J: QUAD, L = 0.5*TU2J, K1 =-KU2J;  QUN2J: QUAD, L = 0.5*TU2J, K1 = KU2J
QUX3J: QUAD, L = 0.5*TU3J, K1 = KU3J;  QUN3J: QUAD, L = 0.5*TU3J, K1 =-KU3J
QUX4J: QUAD, L = TU4J,     K1 = KU4J;  QUN4J: QUAD, L = TU4J,     K1 =-KU4J
MU1J: MARK; MU2J: MARK; MU3J: MARK; MU4J: MARK
QUF1J:  LINE = (QUX1J,MU1J,QUX1J);     QUD1J:  LINE = (QUN1J,MU1J,QUN1J)
QUD2J:  LINE = (QUX2J,MU2J,QUX2J);     QUF2J:  LINE = (QUN2J,MU2J,QUN2J)
QUF3J:  LINE = (QUX3J,MU3J,QUX3J);     QUD3J:  LINE = (QUN3J,MU3J,QUN3J)
QUF4J:  LINE = (QUX4J,MU4J);           QUD4J:  LINE = (MU4J,QUN4J)

!**********************************
!SEXTUPOLES

!60 degrees
!PARA, K2LSF = 0.01; PARA, K2LSD = -0.02
PARA, K2LSF = 0.0145589; PARA, K2LSD = -0.0233231
SF60: SEXTUPOLE, L=LSEXT, K2 = K2LSF / LSEXT
SD60: SEXTUPOLE, L=LSEXT, K2 = K2LSD / LSEXT
SPF6: LINE = (DEND,SF60,DEND);     SPD6: LINE = (DEND,SD60,DEND)

!**********************************
!CELLS AND HALF CELLS

!60 degrees cells, with bends and sextupoles
BD6:    LINE = (5*B,SPD6);     BF6:   LINE = (5*B,SPF6)
CFD6:  LINE = (QFX6,BD6,QDN6);   CDF6:  LINE = (QDX6,BF6,QFN6)
CELLF6:  LINE = (CFD6,MQD6,CDF6)
!60 degree cells, with bends and no sextupoles (B => BEND)
CFDB6: LINE = (QFX6,BVSF6,QDN6);   CDFB6: LINE = (QDX6,BVSF6,QFN6)
CELLFB6: LINE = (CFDB6,MQD6,CDFB6);  CELLDB6: LINE = (CDFB6,MQF6,CFDB6)
!60 degree cells, no bends or sextupoles (E => EMPTY)
CFDE6: LINE = (QFX6,DVSE6,QDN6);  CDFE6: LINE = (QDX6,DVSE6,QFN6)
CELLFE6: LINE = (CFDE6,MQD6,CDFE6);  CELLDE6: LINE = (CDFE6,MQF6,CFDE6)
!60 degree vertical steps ( U => UP, D => DOWN)
CFDU16: LINE = (QFX6,VSU16,QDN6);   CFDD16: LINE = (QFX6,VSD16,QDN6)
CFDU26: LINE = (QFX6,VSU26,QDN6);   CFDD26: LINE = (QFX6,VSD26,QDN6)
!90 degrees separator cell
CFD9:  LINE = (QFX9,DVSE9,QDN9);   CDF9:  LINE = (QDX9,DVSE9,QFN9)
CELLF9:  LINE = (CFD9,MQD9,CDF9)
!90 degree separator cells for horizontal dispersion (H => HORIZONTAL)
CELLFH9:  LINE = (QFX9,HDSN9,QD9,HDSX9,QFN9)
CELLDH9:  LINE = (QDX9,HDSN9,QF9,HDSX9,QDN9)
```

```
!90 degree vertical steps ( U => UP, D => DOWN)
CFDU19: LINE = (QFX9,VSU19,QDN9);    CFDD19: LINE = (QFX9,VSD19,QDN9)
CFDU29: LINE = (QFX9,VSU29,QDN9);    CFDD29: LINE = (QFX9,VSD29,QDN9)
!neutral phase trombone cell
CFDPT:  LINE = (QFXPT,BPT,QDNPT);    CDFPT:  LINE = (QDXPT,BPT,QFNPT)
CELLFPT:  LINE = (CFDPT,MQDPT,CDFPT); CELLDPT:  LINE = (CDFPT,MQFPT,CFDPT)
!join between utilities
CFDEJ:    LINE = (QFXJ,DEJ,QDNJ);     CDFEJ:   LINE = (QDXJ,DEJ,QFNJ)
CELLFEJ:  LINE = (CFDEJ,MQDJ,CDFEJ); CELLDEJ:  LINE = (CDFEJ,MQFJ,CFDEJ)
CDF4J:    LINE = (QDXJ,B4J,QFNJ)
CFD5NJ:   LINE = (QFXJ,B5NJ,QDNJ);  CFD5XJ:  LINE = (QFXJ,B5XJ,QDNJ)

!**********************************
!LINES

!60 degree separators
SEPXU6: LINE = (CELLDE6,MQD6,CELLDB6,MQD6,CDFE6,MQF6,CFDU26)
SEPXD6: LINE = (CELLDE6,MQD6,CELLDB6,MQD6,CDFE6,MQF6,CFDD26)
SEPNU6: LINE = (CFDU26,MQD6,CDFE6,MQF6,CELLFB6,MQF6,CELLFE6)
SEPND6: LINE = (CFDD26,MQD6,CDFE6,MQF6,CELLFB6,MQF6,CELLFE6)
!90 degree separators
SEPXU9: LINE = (CDF9,MQF9,CELLFH9,MQF9,CFDU29)
SEPXD9: LINE = (CDF9,MQF9,CELLFH9,MQF9,CFDD29)
SEPNU9: LINE = (CFDU29,MQD9,CELLDH9,MQD9,CDF9)
SEPND9: LINE = (CFDD29,MQD9,CELLDH9,MQD9,CDF9)
!60 degree low betas
LBXU6: LINE = (10*D1,QX16,D2,QX26,D3,QX36,10*D46,QX46,D56,QFN6,MQF6,CFDU16)
LBXD6: LINE = (10*D1,QX16,D2,QX26,D3,QX36,10*D46,QX46,D56,QFN6,MQF6,CFDD16)
LBNU6: LINE = (CFDU16,MQD6,QDX6,D56,QN46,10*D46,QN36,D3,QN26,D2,QN16,10*D1)
LBND6: LINE = (CFDD16,MQD6,QDX6,D56,QN46,10*D46,QN36,D3,QN26,D2,QN16,10*D1)
!90 degree low betas
LBXU9: LINE = (10*D1,QX19,D2,QX29,D3,QX39,10*D49,QX49,D59,QFN9,MQF9,CFDU19)
LBXD9: LINE = (10*D1,QX19,D2,QX29,D3,QX39,10*D49,QX49,D59,QFN9,MQF9,CFDD19)
LBNU9: LINE = (CFDU19,MQD9,QDX9,D59,QN49,10*D49,QN39,D3,QN29,D2,QN19,10*D1)
LBND9: LINE = (CFDD19,MQD9,QDX9,D59,QN49,10*D49,QN39,D3,QN29,D2,QN19,10*D1)
!join low beta
LBXUJ: LINE = (10*D1,QX1J,D2,QX2J,D3,QX3J,10*D49,QX4J,D59,QFN9,MQF9,CFDU19)
LBXDJ: LINE = (10*D1,QX1J,D2,QX2J,D3,QX3J,10*D49,QX4J,D59,QFN9,MQF9,CFDD19)
LBNUJ: LINE = (CFDU19,MQD9,QDX9,D59,QN4J,10*D49,QN3J,D3,QN2J,D2,QN1J,10*D1)
LBNDJ: LINE = (CFDD19,MQD9,QDX9,D59,QN4J,10*D49,QN3J,D3,QN2J,D2,QN1J,10*D1)

!trombones
TROM: LINE=(PTD1,BPT,PTF2,BPT,PTD3,BPT,PTF4,BPT,PTD5,BPT,&
PTF6,BPT,PTD7,BPT,PTF8)
CONSTROM: LINE=(CTD1,BPT,CTF2,BPT,CTD3,BPT,CTF4,BPT,CTD5,BPT,&
CTF6,BPT,CTD7,BPT,CTF8)
!join between utilities
JFDB:   LINE = (CFD5NJ,MQDJ,CDF4J,MQFJ,CFD5XJ)
JOIN:   LINE = (JFDB,MQDJ,CDFEJ,MQFJ,JFDB)
!90 degree utilities
UTILX9:  LINE = (10*DU19,QUF19,DU29,QUD29,10*DU39,QUF39,DU49,QUF49)
UTILN9:  LINE = (QUD49,DU49,QUD39,10*DU39,QUF29,DU29,QUD19,10*DU19)
UTX9:   LINE = (UTILX9,MQF9,CELLFH9,MQF9,CFD9)
UTN9:   LINE = (CFD9,MQD9,CELLDH9,MQD9,UTILN9)
!60 degree utilites
UTILX6:  LINE = (10*DU16,QUF16,DU26,QUD26,10*DU36,QUF36,DU46,QUF46,DU56,QDN6)
```

```
UTILN6:  LINE = (QFX6,DU56,QUD46,DU46,QUD36,10*DU36,QUF26,DU26,QUD16,10*DU16)
UTX6:    LINE = (UTILX6,MQD6,CELLDB6,MQD6,CELLDE6)
UTN6:    LINE = (CELLFE6,MQF6,CELLFB6,MQF6,UTILN6)
!join utilities
UTILXJ:  LINE = (10*DU1J,QUF1J,DU2J,QUD2J,10*DU3J,QUF3J,DU4J,QUF4J)
UTILNJ:  LINE = (QUD4J,DU4J,QUD3J,10*DU3J,QUF2J,DU2J,QUD1J,10*DU1J)
JUTNJ:  LINE = (JOIN,MQDJ,UTILNJ)

!**********************************
!ASSEMBLIES

MIP1: MARK; MIP2: MARK; MIP3: MARK; MIP4: MARK; MIP5: MARK; MIP6: MARK
MUIP1: MARK; MUIP2: MARK
X1N: LINE = (SEPND6,LBND6);      X1X: LINE = (LBXD9,SEPXD9)
X1:  LINE = (X1N,MIP1,X1X)
X2:  LINE = (SEPNU9,LBNU9,MIP2,LBXU9,SEPXU9)
X3:  LINE = (SEPND9,LBND9,MIP3,LBXD9,SEPXD9)
X4:  LINE = (SEPNU9,LBNU9,MIP4,LBXU6,SEPXU6)
ARC: LINE = (CDF6,MQF6,180*CELLF6)
!ARC: LINE = (CDF6)
X5:  LINE = (SEPND6,LBND6,MIP5,LBXD9,SEPXD9)
X6:  LINE = (SEPNU9,LBNU9,MIP6,LBXUJ,SEPXU9)
U1:  LINE = (UTN9,MUIP1,UTILXJ)
U2:  LINE = (UTILNJ,MUIP2,UTX6)
NORTH: LINE = (X1X,TROM,X2,TROM,X3,TROM,X4)
SOUTH: LINE = (X5,TROM,X6,CONSTROM,U1,JOIN,U2)
WHOLE: LINE = (MIP1,NORTH,ARC,SOUTH,ARC,X1N)
```